

Домашна 3 – 211177

За оваа домашна ги скокнав сите опционални барања односно дирек отидов на Дел3 барање 3 заедно споено со барање 4

```
USE WideWorldImportersDW;
GO

DROP TABLE IF EXISTS [Integration].[ETL_Status]

CREATE TABLE [Integration].[ETL_Status] (
    [StatusID] INT IDENTITY(1,1) PRIMARY KEY,
    [TableName] NVARCHAR(128) NOT NULL,
    [StartTime] DATETIME NOT NULL,
    [EndTime] DATETIME NULL,
    [Status] NVARCHAR(50) NOT NULL,
    [RowsAffected] INT NULL,
    [ErrorMessage] NVARCHAR(MAX) NULL,
    [LineageKey] INT NOT NULL DEFAULT 1
);

CREATE NONCLUSTERED INDEX [IX_ETL_Status_TableName_StartTime]
ON [Integration].[ETL_Status] ([TableName], [StartTime]);
```

```
CREATE OR ALTER PROCEDURE [Integration].[usp_Load_Fact_Purchase]
AS
BEGIN
    SET NOCOUNT ON;

    DECLARE @StatusID INT;
    DECLARE @RowsAffected INT = 0;
    DECLARE @ErrorMessage NVARCHAR(MAX);
    DECLARE @StartTime DATETIME = GETDATE();
    DECLARE @LineageKey INT = 1;
    DECLARE @LastRunTime DATETIME;

    SELECT @LastRunTime = ISNULL(MAX([StartTime]), '1900-01-01')
    FROM [Integration].[ETL_Status]
    WHERE [TableName] = 'Fact.Purchase' AND [Status] = 'Completed';

    INSERT INTO [Integration].[ETL_Status] (
        [TableName], [StartTime], [Status], [LineageKey]
    )
    VALUES (
        'Fact.Purchase', @StartTime, 'Started', @LineageKey
    );

    SET @StatusID = SCOPE_IDENTITY();

    BEGIN TRY
```

```

IF OBJECT_ID('tempdb..#PurchaseStaging') IS NOT NULL
    DROP TABLE #PurchaseStaging;

CREATE TABLE #PurchaseStaging (
    [DateKey] DATE NOT NULL,
    [SupplierKey] INT NOT NULL,
    [StockItemKey] INT NOT NULL,
    [WWIPurchaseOrderID] INT NULL,
    [OrderedOuters] INT NOT NULL,
    [OrderedQuantity] INT NOT NULL,
    [ReceivedOuters] INT NOT NULL,
    [Package] NVARCHAR(50) NOT NULL,
    [IsOrderFinalized] BIT NOT NULL,
    [LineageKey] INT NOT NULL
);

INSERT INTO #PurchaseStaging (
    [DateKey], [SupplierKey], [StockItemKey], [WWIPurchaseOrderID],
    [OrderedOuters], [OrderedQuantity], [ReceivedOuters], [Package],
    [IsOrderFinalized], [LineageKey]
)
SELECT
    COALESCE(CAST(p.OrderDate AS DATE), '1900-01-01') AS [DateKey],
    s.[Supplier Key],
    si.[Stock Item Key],
    p.[PurchaseOrderID] AS [WWIPurchaseOrderID],
    pl.[OrderedOuters],
    pl.[OrderedOuters] * si.[Quantity Per Outer] AS [OrderedQuantity],
    pl.[ReceivedOuters],
    ISNULL(pt.[PackageName], 'Unknown') AS [Package],
    CASE WHEN p.[IsOrderFinalized] = 1 THEN 1 ELSE 0 END AS [IsOrderFinalized],
    @LineageKey
FROM [WideWorldImporters].[Purchasing].[PurchaseOrders] p
INNER JOIN [WideWorldImporters].[Purchasing].[PurchaseOrderLines] pl
    ON p.[PurchaseOrderID] = pl.[PurchaseOrderID]
INNER JOIN [WideWorldImportersDW].[Dimension].[Supplier] s
    ON p.[SupplierID] = s.[WWI Supplier ID]
INNER JOIN [WideWorldImportersDW].[Dimension].[Stock Item] si
    ON pl.[StockItemID] = si.[WWI Stock Item ID]
INNER JOIN [WideWorldImportersDW].[Dimension].[Date] d
    ON COALESCE(CAST(p.OrderDate AS DATE), '1900-01-01') = d.[Date]
LEFT JOIN [WideWorldImporters].[Warehouse].[PackageTypes] pt
    ON pl.[PackageTypeID] = pt.[PackageTypeID]
WHERE (p.[LastEditedWhen] > @LastRunTime OR pl.[LastEditedWhen] > @LastRunTime);

INSERT INTO [WideWorldImportersDW].[Fact].[Purchase] (
    [Date Key], [Supplier Key], [Stock Item Key], [WWI Purchase Order ID],
    [Ordered Outers], [Ordered Quantity], [Received Outers], [Package],
    [Is Order Finalized], [Lineage Key]
)
SELECT
    [DateKey], [SupplierKey], [StockItemKey], [WWIPurchaseOrderID],
    [OrderedOuters], [OrderedQuantity], [ReceivedOuters], [Package],
    [IsOrderFinalized], [LineageKey]

```

```

FROM #PurchaseStaging;

SET @RowsAffected = @@ROWCOUNT;

UPDATE [Integration].[ETL_Status]
SET
    [EndTime] = GETDATE(),
    [Status] = 'Completed',
    [RowsAffected] = @RowsAffected,
    [ErrorMessage] = NULL
WHERE [StatusID] = @StatusID;

DROP TABLE #PurchaseStaging;
END TRY
BEGIN CATCH
    SET @ErrorMessage = ERROR_MESSAGE();

    UPDATE [Integration].[ETL_Status]
    SET
        [EndTime] = GETDATE(),
        [Status] = 'Failed',
        [RowsAffected] = @RowsAffected,
        [ErrorMessage] = @ErrorMessage
    WHERE [StatusID] = @StatusID;

    IF OBJECT_ID('tempdb..#PurchaseStaging') IS NOT NULL
        DROP TABLE #PurchaseStaging;

    THROW;
END CATCH;
END;
GO

```

```

EXEC [Integration].[usp_Load_Fact_Purchase];
SELECT
    [StatusID],
    [TableName],
    [StartTime],
    [EndTime],
    [Status],
    [RowsAffected],
    [ErrorMessage],
    [LineageKey]
FROM [WideWorldImportersDW].[Integration].[ETL_Status]
WHERE [TableName] = 'Fact.Purchase'
ORDER BY [StartTime] DESC;

```

SQL

(2 rows affected)

Total execution time: 00:00:00.002

	StatusID	TableName	StartTime	EndTime	Status	RowsAffected	ErrorMessage	LineageKey
1	2	Fact.Purchase	2025-04-29 15:14:28.270	2025-04-29 15:14:28.290	Completed	0	NULL	1
2	1	Fact.Purchase	2025-04-29 15:12:50.653	2025-04-29 15:12:51.250	Completed	38722	NULL	1

1

SELECT TOP 10 * FROM Fact.Purchase

SQL

(10 rows affected)

Total execution time: 00:00:00.016

	Purchase Key	Date Key	Supplier Key	Stock Item Key	MWI Purchase Order ID	Ordered Outers	Ordered Quantity	Received Outers	Package	Is Order Finalized
1	5	2013-01-01	17	143	2	1	12	1	Carton	1
2	95	2013-01-02	17	143	7	6	72	6	Carton	1
3	140	2013-01-03	17	143	12	11	132	11	Carton	1

PROBLEMS

23

OUTPUT

TERMINAL

TASKS

✓

Restore Database succeeded

localhost | WideWorldImporters

2:03:06 PM - 2:03:59 PM (00:00:53)

✓

Restore Database succeeded

localhost | WideWorldImportersDW

2:02:55 PM - 2:03:54 PM (00:00:58)